

## Inhaltsverzeichnis

1	Motivation.....	2
2	Der "übliche" Weg.....	2
2.1	Video soll interlaced bleiben.....	2
2.1.1	Ohne temporale Filter (z.B. nur spatiale Rauschfilter, Resizen u. Farbanpassungen) .....	2
2.1.2	Mit temporalen Filtern (z.B. temporale Rauschfilter) .....	2
2.2	Video soll deinterlaced (progressive) werden .....	3
3	Filtern von interlaced Video mittels Bobber .....	3
3.1	Verwendete Funktionen.....	3
3.2	Video soll interlaced bleiben.....	5
3.3	Video soll deinterlaced (progressive) werden .....	5
4	Filtern mit Depan .....	5
5	Resizing .....	5
6	Danksagung.....	6

## 1 Motivation

Der "übliche" Weg interlaced Video in AviSynth zu filtern ist in Kapitel 2 *Der "übliche" Weg* dargestellt. Unter Verwendung eines Bobbers können jedoch z.T. wesentlich bessere Ergebnisse erzielt werden. V.a. können temporale Rauschfilter wesentlich effektiver und mit geringeren Thresholds eingesetzt werden. In Kapitel 3 *Filtern von interlaced Video mittels Bobber* werden je nach gewünschtem Endergebnis (interlaced od. progressive) und der verwendeten Filter (nur spatial od. auch temporal) die jeweiligen Wege gezeigt.

In untenstehenden Beispielen wird eine Quelldatei mit Auflösung 704 x 576 gefiltert und in eine Zieldatei mit Auflösung 352 x 576 (1/2 DVD) umgewandelt. Zum Resizing siehe auch Kap. 5.

## 2 Der "übliche" Weg

### 2.1 Video soll interlaced bleiben

#### 2.1.1 Ohne temporale Filter (z.B. nur spatiale Rauschfilter, Resizen u. Farbanpassungen)

```
AviSource("InterlacedVideo.avi",true,"YUY2")

Crop(6,12,-4,-16)

SeparateFields()

UnDot()                                # nur spatiale Rauschfilter
Lanczos4Resize(348,274)
ColorYUV(levels="PC->TV")
FadeOut(50)

Weave()

AddBorders(2,12,2,16)
```

#### 2.1.2 Mit temporalen Filtern (z.B. temporale Rauschfilter)

```
AviSource("InterlacedVideo.avi",true,"YUY2")

Crop(6,12,-4,-16)

SeparateFields()

UnDot()                                # spatiale Rauschfilter

o = SelectOdd().TemporalCleaner(3,6) # temporale Rauschfilter
e = SelectEven().TemporalCleaner(3,6) # temporale Rauschfilter
Interleave(e,o)

# spatiale Rauschfilter auch hier möglich

Lanczos4Resize(348,274)
ColorYUV(levels="PC->TV")
FadeOut(50)

Weave()

AddBorders(2,12,2,16)
```

## 2.2 Video soll deinterlaced (progressive) werden

```
AviSource("InterlacedVideo.avi",true,"YUY2")

AssumeTFF()                # ggf. AssumeBFF()

Crop(6,12,-4,-16)

TDeint()                   # Deinterlacen
#TomsMoComp(-1,5,0)       # Alternative zu TDeint

UnDot().TemporalCleaner(3,6)
Lanczos4Resize(348,548)
ColorYUV(levels="PC->TV")
AddBorders(2,12,2,16)
FadeOut(50)
```

## 3 Filtern von interlaced Video mittels Bobber

Sollen als Rauschfilter ausschließlich spatiale Filter zum Einsatz kommen, so ist Variante 2.1.1 bzw. 2.2 ohne weiteres ausreichend. Sollen jedoch temporale Filter eingesetzt werden, so sollte man neben 2.1.2 und 2.2 auch die folgenden Varianten testen:

Durch Einsatz von Bobbern erhalten die temporalen Filter die doppelte Bewegungsinformation und können dadurch effektiver und i.d.R. auch mit kleineren Thresholds als sonst eingesetzt werden.

Zusätzlich lässt sich die Effektivität der Rauschfilter durch Einsatz von Depan (siehe 4) noch weiter steigern.

### 3.1 Verwendete Funktionen

```
LoadPlugin("....\TDeint.dll")          # mind. TDeint 0.9.7
LoadPlugin("....\TomsMoComp.dll")

#####
# TDeintBob(clip a, bool "tff")
#
# Erzeugt aus interlaced Video ein progressives doppelter Framerate
# (z.B. aus 25 fps interlaced ein 50 fps progressive)
#
# tff gibt die Field order des clip a an:
# true --> Top Field First (Upper Field First)
# false --> Bottom Field First (Lower Field First)
# Falls nicht angegeben, wird true (Top Field First) angenommen
#####

#####
# TMCBob(clip a, bool "tff")
#
# Erzeugt aus interlaced Video ein progressives doppelter Framerate
# (z.B. aus 25 fps interlaced ein 50 fps progressive)
#
# tff gibt die Field order des clip a an:
# true --> Top Field First (Upper Field First)
# false --> Bottom Field First (Lower Field First)
# Falls nicht angegeben, wird true (Top Field First) angenommen
#####
```

# Filtern von interlaced Video in AviSynth

grua

V4.00

```
#####
# ReinterlaceBob(clip a, bool "tff")
#
# Erzeugt aus einem progressiven ein interlaced Video halber Framerate
# (z.B. aus 50 fps progressive ein 25 fps interlaced)
#
# tff gibt die Field order des zu erzeugenden interlaced Videos an:
# true --> Top Field First (Upper Field First)
# false --> Botttom Field First (Lower Field First)
# Falls nicht angegeben, wird true (Top Field First) angenommen
#####

#####
# HalfFpsBob(clip a)
#
# Erzeugt aus einem via TDeintBob auf doppelte Framerate gebobbtten Video
# ein progressives halber Framerate
# (z.B. aus 50 fps progressive ein 25 fps progressive)
#####

function TDeintBob(clip a, bool "tff") {
    tff = Default(tff,true)
    a = tff ? a.AssumeTFF() : a.AssumeBFF()           1)
    a.TDeint(mode=1)
}

function TMCBob(clip a, bool "tff") {
    tff = Default(tff,true)
    a = tff ? a.AssumeTFF() : a.AssumeBFF()           1)
    f = a.TomsMoComp(-1,5,0)
    e = a.DoubleWeave().SelectOdd().TomsMoComp(-1,5,0)
    Interleave(f,e).AssumeFrameBased()
}

function ReinterlaceBob(clip a, bool "tff") {
    tff = Default(tff,true)
    a = tff ? a.AssumeTFF() : a.AssumeBFF()           2)
    a = a.ConvertToYUY2()
    SeparateFields(a).SelectEvery(4,0,3).Weave()
}

function HalfFpsBob(clip a) {
    SelectOdd(a)
}
}
```

mvBob() siehe:

<http://forum.doom9.org/showthread.php?s=&postid=606487#post606487>

<sup>1)</sup> AviSource() setzt immer auf BFF, auch wenn das AVI tats. TFF ist. Daher wird ggf. wieder auf TFF korrigiert.

<sup>2)</sup> Bob() und AssumeFrameBased() setzen intern immer auf BFF, auch wenn die Quelle tats. TFF ist. Daher wird ggf. wieder auf TFF korrigiert.

## 3.2 Video soll interlaced bleiben

```
AviSource("InterlacedVideo.avi",true,"YUY2")

# Einen der folgenden drei Bobber verwenden:
AssumeTFF().mvBob()           # bzw. AssumeBFF().mvBob()
#TDeintBob()                  # bzw. TDeintBob(tff=false)
#TMCBob()                      # bzw. TMCBob(tff=false)

Crop(6,12,-4,-16)

UnDot().TemporalCleaner(3,6)
Lanczos4Resize(348,548)
ColorYUV(levels="PC->TV")
AddBorders(2,12,2,16)
FadeOut(100)

ReinterlaceBob()              # bzw. ReinterlaceBob(tff=false)
```

## 3.3 Video soll deinterlaced (progressive) werden

Alles ist vollständig identisch zu 3.2 durchzuführen, lediglich die letzte Zeile (ReinterlaceBob) ist durch diese zu ersetzen:

```
HalfFpsBob()
```

## 4 Filtern mit Depan

Grundsätzlich ist es zu empfehlen temporale Rauschfilter immer in Verbindung mit Depan einzusetzen, da die temporalen Filter dann wesentlich mehr Informationen zum Denoisen zur Verfügung haben.

Aus

```
temporalerRauschfilter()
```

wird dann

```
data=depanestimate(last,pixaspect=1.094)
depaninterleave(last,data,pixaspect=1.094,prev=2,next=2)
temporalerRauschfilter()
SelectEvery(5,2)
```

Dies benötigt aber ein mehrfaches der ursprünglichen Rechenzeit! Ein Kompromiss aus Qualität u. Rechenzeit wäre z.B.:

```
data=depanestimate(last,pixaspect=1.094)
depaninterleave(last,data,pixaspect=1.094)
temporalerRauschfilter()
SelectEvery(3,1)
```

## 5 Resizing

Betreffend Resizing empfehle ich um höchstmögliche Qualität zu erreichen vertikal nicht zu resizen. Falls unsaubere Ränder gecroppt werden, so sollte man diese nach dem Filtern durch AddBorders wieder durch schwarze Ränder ersetzen. Wenn Resizing, dann nur horizontal! So wurde dies auch in obigen Beispielen durchgeführt.

## 6 Danksagung

Obiges Vorgehen entstammt den Vorschlägen von scharfis\_brain (<http://forum.gleitz.info>) bei welchem ich mich dafür herzlich bedanken möchte!